Solving Constraint Satisfaction and Optimization Problems with CLP(FD)

http://www.probp.com/clpfd_tutorial.pdf

Neng-Fa Zhou
The City University of New York
zhou@sci.brooklyn.cuny.edu

Outline of Lectures

- Constraint Satisfaction Problems (CSPs)
- Constraint programming by example
  - Rabbit and chiken
  - Kakuro
  - Knapsack
  - Magic square
  - Graph-coloring
  - N-Queens
  - Traveling salesman
  - Maximum flow
  - Scheduling
  - Planning
  - Routing
  - Protein structure predication
- Constraint propagation algorithms and their implementation

Requirements and Declaimers

- CLP(FD) systems
  - B-Prolog (www.probp.com)
  - Other systems: SICStus & ECLiPSe
- References (for example),
  - www.constraint.org
  - http://kti.mff.cuni.cz/~bartak/constraints/
  - *Programming with Constraints*, by Marriott and Stuckey, The MIT Press, 1999
- Only CLP(FD) is covered
  - No constraint programming over other domains
  - No constraint libraries such as Ilog and Gcode

Constraint Satisfaction Problems

- CSP
  - *A set of variables $V=\{V_1,\ldots,V_n\}$*
  - *Each variable has a domain $V_i :: D_i$*
  - *A set of constraints*
- *Example*
  - A:{0,1}, B:{0,1}, C:{0,1}
  - C = A and B
- Solution to CSP
  - *An assignment of values to the variables that satisfies all the constraints*

CLP(FD) by Example (I)

- The rabbit and chicken problem
- The Kakuro puzzle
- The knapsack problem
- Exercises

The Rabbit and Chicken Problem

- In a farmyard, there are only chickens and rabbits. Its is known that there are 18 heads and 58 feet. How many chickens and rabbits are there?

```
go:-
        [X,Y] :: 1..58,
        X+Y #= 18,
        2*X+4*Y #= 58,
        labeling([X,Y]),
        writeln([X,Y]).
```

Break the Code Down

```
go:-
        [X,Y] :: 1..58,
        X+Y #= 18,
        2*X+4*Y #= 58,
        labeling([X,Y]),
        writeln([X,Y]).
```

- `go` -- a predicate
- `X,Y` -- variables
- `1..58` -- a domain
- `X :: D` -- a domain declaration
- `E1 #= E2` -- equation (or equality constraint)
- `labeling(Vars)--` find a valuation for variables that satisfies the constraints
- `writeln(T)` -- a Prolog built-in

Running the Program

- Save the program into a file named *rabbit.pl*
- Start B-Prolog from the directory where the file is located

```
| ?- cl(rabbit)
Compiling::rabbit.pl
compiled in 0 milliseconds
loading::rabbit.out

yes
| ?- go
[7,11]
```

The Kakuro Puzzle

- Kakuro, another puzzle originated in Japan after Sudoku, is a mathematical version of a crossword puzzle that uses sums of digits instead of words. The objective of Kakuro is to fill in the white squares with digits such that each down and across “word” has the given sum. No digit can be used more than once in each “word”.

An Example



A Kakuro puzzle

```
go:-
    Vars=[X1,X2,…,X16],
    Vars :: 1..9,
    word([X1,X2],5),
    word([X3,X4,X5,X6],17),
    …
    word([X10,X14],3),
    labeling(Vars),
    writeln(Vars).

word(L,Sum):-
    sum(L) #= Sum,
    all_different(L).
```

Break the Code Down

- `sum(L) #= Sum`
  The sum of the elements in `L` makes `Sum`.

  e.g., `sum([X1,X2,X3]) #= Y` is the same as `X1+X2+X3 #= Y`.

- `all_different(L)`
  Every element in `L` is different.

The Knapsack Problem

A smuggler has a knapsack of 9 units. He can smuggle in bottles of whiskey of size 4 units, bottles of perfume of size 3 units, and cartons of cigarettes of size 2 units. The profit of smuggling a bottle of whiskey, a bottle of perfume or a carton of cigarettes is 15, 10 and 7, respectively. If the smuggler will only take a trip, how can he take to make the largest profit?

```
go:-
    [W,P,C] :: 0..9,
    4*W+3*P+2*C #=< 9,
    maxof(labeling([W,P,C]),15*W+10*P+7*C),
    writeln([W,P,C]).
```

Break the Code Down

- `maxof(Goal,Exp)`
  Find an instance of `Goal` that is true and maximizes `Exp`.

Exercises

1. Tickets to a carnival cost 250 JPY for students and 400 JPY for adults. If a group buys 10 tickets for a total of 3100 JPY, how many of the tickets are for students?
2. The product of the ages, in years, of three teenagers is 4590. None of the teens are the same age. What are the ages of the teenagers?
3. Suppose that you have 100 pennies, 100 nickels, and 100 dimes. Using at least one coin of each type, select 21 coins that have a total value of exactly $1.00. How many of each type did you select?

Exercises (Cont.)

4. If m and n are positive integers, neither of which is divisible by 10, and if mn = 10,000, find the sum m+n.
5. The arithmetic cryptographic puzzle: Find distinct digits for S, E, N, D, M, O, R, Y such that S and M are non-zero and the equation SEND+MORE=MONEY is satisfied.
6. A magic square of order 3x3 is an arrangement of integers from 1 to 9 such that all rows, all columns, and both diagonals have the same sum.

Exercises (Cont.)

7. Place the numbers 2,3,4,5,6,7,8,9,10 in the boxes so that the sum of the numbers in the boxes of each of the four circles is 27.
8. Sudoku puzzle.



8	6	7			5	9	1	
1				7			8	5
	3							
			7	6	2	1		
	8			9			6	
		2	8	1	4			
							3	
9	1			3				6
	4	3	1			8	2	9

Exercises (Cont.)

9. A factory has four workers *w1*,*w2*,*w3*,*w4* and four products *p1*,*p2*,*p3*,*p4*. The problem is to assign workers to products so that each worker is assigned to one product, each product is assigned to one worker, and the profit maximized. The profit made by each worker working on each product is given in the matrix.

Profit matrix is:

	*p*1	*p*2	*p*3	*p*4
*w*1	7	1	3	4
*w*2	8	2	5	1
*w*3	4	3	7	2
*w*4	3	1	6	3

Review of CLP(FD)

- Declaration of domain variables
  - X :: L..U
  - [X1,X2,...,Xn] :: L..U
- Constraints
  - Exp R Exp (
    - R is one of the following: #=, #¥=, #>, #>=, #<, #=
    - Exp may contain +, -, *, /, //, mod, sum, min, max
  - all_different(L)
- Labeling
  - labeling(L)
  - minof(labeling(L),Exp) and maxof(labeling(L),Exp)
- Prolog built-ins: T1=T2, X is Exp and writeln(T)

CLP(FD) by Example (II)

- The graph coloring problem
- The N-queens problem
- The magic square problem
- Exercises

Graph Coloring

- Given a graph G=(V,E) and a set of colors, assign a color to each vertex in V so that no two adjacent vertices share the same color.



The map of Kyushu

Fukuoka
Kagoshima
Kumamoto
Miyazaki
Nagasaki
Oita
Saga

Color the Map of Kyushu

```
go:-
        Vars=[Cf,Cka,Cku,Cm,Cn,Co,Cs],
        Vars :: [red,blue,purple],
        Cf #¥= Cs,
        Cf #¥= Co,
        …
        labeling(Vars),
        writeln(Vars).
```

- Atoms
  - `red`, `blue`, `purple`

The N-Queens Problem

- Find a layout for the N queens on an NxN chessboard so that no queens attack each other. Two queens attack each other if they are placed in the same row, the same column, or the same diagonal.



Qi: the number of the row for the i^{th} queen.

for each two different variables Qi and Qj

Qi #¥= Qj %not same row

abs(Qi-Qj) #¥= abs(i-j) %not same diagonal

The N-Queens Problem (Cont.)

```
queens(N):-
    length(Qs,N),
    Qs in 1..N,
    foreach((I in 1..N-1, J in I+1..N),
        [Qi,Qj],
        (nth(I,Qs,Qi),
         nth(J,Qs,Qj),
         Qi #¥= Qj,
         abs(Qi-Qj) #¥= abs(I-J))),
     labeling_ff(Qs),
     writeln(Qs).
```

Break the Code Down

- length(L,N)

```
?-length([a,b,c],N)
N = 3
?-length(L,3)
L = [_310,_318,_320]
```

- foreach(Iterators,LocalVars,Goal)

```
?-L=[a,b,c],foreach(E in L, writeln(E))
```

- nth(I,L,E)

```
?-nth(2,[a,b,c],X)
X = b
```

Generating a CLP(FD) Program for Solving N-Queens Problem

```
gen_queen(int n){
  int i,j;
  printf("go:-¥n");
  printf("¥tVars=[");
  for (i=1;i<n;i++) printf("Q%d,",i);
  printf("Q%d],¥n",n);
  printf(“¥tVars :: 1..%d,¥n",n);
  for (i=1;i<n;i++){
    for (j=i+1;j<=n;j++){
      printf("¥tQ%d #¥¥= Q%d,¥n",i,j);
      printf("¥tabs(Q%d-Q%d) #¥¥= %d,¥n",i,j,abs(i-j));
    }
  }
  printf("¥tlabeling_ff(Vars),¥n");
  printf("¥twriteln(Vars).¥n");
}
```

Generated CLP(FD) Program for 4-Queens Problem

```
go:-
    Vars=[Q1,Q2,Q3,Q4],
    Vars :: 1..4,
    Q1 #¥= Q2,
    abs(Q1-Q2) #¥= 1,
    Q1 #¥= Q3,
    abs(Q1-Q3) #¥= 2,
    Q1 #¥= Q4,
    abs(Q1-Q4) #¥= 3,
    Q2 #¥= Q3,
    abs(Q2-Q3) #¥= 1,
    Q2 #¥= Q4,
    abs(Q2-Q4) #¥= 2,
    Q3 #¥= Q4,
    abs(Q3-Q4) #¥= 1,
    labeling_ff(Vars),
    writeln(Vars).
```

Break the Code Down

- `labeling_ff(L)`
  - Label the variables in L by selecting first a variable with the smallest domain. If there are multiple variables with the same smallest domain, then the left-most one is chosen.

N-Queens Problem

```
queens(N):-
        length(List,N),
        List :: 1..N,
        constrain_queens(List),
        labeling_ff(List),
        writeln(List).

constrain_queens([]).
constrain_queens([X|Y]):-
        safe(X,Y,1),
        constrain_queens(Y).
```

```
safe(_,[],_).
safe(X,[Y|T],K):-
        noattack(X,Y,K),
        K1 is K+1,
        safe(X,T,K1).

noattack(X,Y,K):-
        X #¥= Y,
        abs(X-Y) #¥= K.
```

Magic Square

- A magic square of order NxN is an arrangement of integers from 1 to N^2 such that all rows, all columns, and both diagonals have the same sum

X11 X12 … X1n

…

Xn1 Xn2 … Xnn

$$\forall_{i=1\ldots n} \sum_{j=1}^{n} X_{ij} = Sum$$
$$\forall_{j=1\ldots n} \sum_{i=1}^{n} X_{ij} = Sum$$
$$\sum_{i=1}^{n} X_{ii} = Sum$$
$$\sum_{i=1}^{n} X_{i(n-i+1)} = Sum$$

Exercises

1. Write a CLP(FD) program to test if the map of Japan is 3-colorable (can be colored with three colors).
2. Write a program in your favorite language to generate a CLP(FD) program for solving the magic square problem.

Exercises (Cont.)

3. Find an integer programming problem and convert it into CLP(FD).
4. Find a constraint satisfaction or optimization problem and write a CLP(FD) program to solve it.

CLP(Boolean): A Special Case of CLP(FD)

```
<BooleanExpression> ::=
     0 |                                                    /* false */
     1 |                                                    /* true */
     <Variable> |
     <Expression> #= <Expression> |
     <Expression> #\= <Expression> |
     <Expression> #> <Expression> |
     <Expression> #>= <Expression> |
     <Expression> #< <Expression> |
     <Expression> #=< <Expression> |
     #\ <BooleanExpression> |                               /* not */
     <BooleanExpression> #/\ <BooleanExpression> |          /* and */
     <BooleanExpression> #\/ <BooleanExpression> |          /* or */
     <BooleanExpression> #=> <BooleanExpression> |          /* imply */
     <BooleanExpression> #<=> <BooleanExpression> |         /* equivalent */
     <BooleanExpression> #\ <BooleanExpression>             /* xor */
```

CLP(FD) by Example (III)

- Maximum flow
- Scheduling
- Traveling salesman problem (TSP)
- Planning
- Routing
- Protein structure predication

Maximum Flow Problem

- Given a network G=(N,A) where N is a set of nodes and A is a set of arcs. Each arc (i,j) in A has a capacity Cij which limits the amount of flow that can be sent throw it. Find the maximum flow that can be sent between a single source and a single sink.

Maximum Flow Problem (Cont.)



Capacity matrix

	1	2	3	4	5	6	7	8
1		3	2	3				
2							5	
3		1				1		
4			2		2			
5								5
6		4			2			1
7						2		3
8								

Maximum Flow Problem (Cont.)

```
go:-
    Vars=[X12,X13,X14,X27,X32,X36,X43,
          X45,X58,X62,X65,X68,X76,X78],
    X12 :: 0..3, X13 :: 0..2, X14 :: 0..3,
    X27 :: 0..5, X32 :: 0..1, X36 :: 0..1,
    X43 :: 0..2, X45 :: 0..2, X58 :: 0..5,
    X62 :: 0..4, X65 :: 0..5, X68 :: 0..1,
    X76 :: 0..2, X78 :: 0..3,
    X12+X32+X62-X27 #= 0,
    X13+X43-X32-X36 #= 0,
    X14-X43-X45 #= 0,
    X45+X65-X58 #= 0,
    X36+X76-X62-X65-X68 #= 0,
    X27-X76-X78 #= 0,
    Max #= X58+X68+X78,
    maxof(labeling(Vars),Max),
    writeln(sol(Vars,Max)).
```

Scheduling Problem

- Four roommates are subscribing to four newspapers. The following gives the amounts of time each person spend on each newspaper:

Person/Newspaper/Minutes

==========

Person	Asahi	Nishi	Orient	Sankei
Akiko	60	30	2	5
Bobby	75	3	15	10
Cho	5	15	10	30
Dola	90	1	1	1

Akiko gets up at 7:00, Bobby gets up at 7:15, Cho gets up at 7:15, and Dola gets up at 8:00. Nobody can read more than one newspaper at a time and at any time a newspaper can be read by only one person. Schedule the newspapers such that the four persons finish the newspapers at an earliest possible time.

Scheduling Problem (Cont.)

- Variables
  - For each activity, a variable is used to represent the start time and another variable is used to represent the end time.
    - A_Asahi : The start time for Akiko to read Asahi
    - EA_Asahi: The time when Akiko finishes reading Asahi
- Constraints
  - A_Asahi #>= 7*60 : Akiko gets up at 7:00
  - Nobody can read more than one newspaper at a time
  - A newspaper can be read by only one person at a time
- The objective function
  - Minimize the maximum end time

Scheduling Problem (Cont.)

```
go:-
    Vars = [A_Asahi,A_Nishi,A_Orient,A_Sankei,…],
    A_Asahi #>= 7*60, A_Nishi #>= 7*60, …
    B_Asahi #>=7*60+15, B_Nishi #>= 7*60+15, …
    …
    cumulative([A_Asahi,A_Nishi,A_Orient,A_Sankei],
               [60,30,2,5],[1,1,1,1],1),
    …
    EA_Asahi #= A_Asahi+60, EA_Nishi #= A_Nishi+30,
    …
    max([EA_Asahi,EA_Nishi,…],Max),
    minof(labeling(Vars),Max),
    writeln(Vars).
```

Break the Code Down

- `cumulative(Starts,Durations,Resources,Limit)`

  Let `Starts` be [S1,S2,...,Sn], `Durations` be [D1,D2,...,Dn] and `Resources` be [R1,R2,...,Rn]. For each job i, Si represents the start time, Di the duration, and Ri the units of resources needed. `Limit` is the units of resources available at any time.

  The jobs are mutually disjoint when `Resources` is [1,…,1] and `Limit` is 1.

  Si #>= Sj+Dj #¥/ Sj #>= Si+Di (for i,j=1..n, i≠ j)

Traveling Salesman Problem

- Given an undirected graph G=(V,E), where V is the set of nodes and E the set of edges, each of which is associated with a positive integer indicating the distance between the two nodes, find a shortest possible Hamiltonian cycle that connects all the nodes.

Traveling Salesman Problem (Cont.)

```
go:-
    max_node_num(N), % Nodes are numbered 1,2, …, N
    length(Vars,N),
    decl_domains(Vars,1),
    circuit(Vars),
    findall(edge(X,Y,W),edge(X,Y,W),Edges),
    collect_weights(Edges,Vars,Weights),
    TotalWeight #= sum(Weights),
    minof(labeling_ff(Vars),TotalWeight,writeln((Vars,TotalWeight))).

decl_domains([],_).
decl_domains([Var|Vars],X):-
    findall(Y,edge(X,Y,_),Ys),
    Var :: Ys,
    X1 is X+1,
    decl_domains(Vars,X1).

collect_weights([],_,[]).
collect_weights([edge(X,Y,W)|Es],Vars,[B*W|Ws]):-
    nth(X,Vars,NX),
    nth(Y,Vars,NY),
    B #<=> (NX#=Y #¥/ NY#=X),
    collect_weights(Es,Vars,Ws).
```

Break the Code Down

- `circuit(L)`
  Let `L=[X1,X2,…,Xn]`. A valuation satisfies the constraint if `1->X1,2->X2, …, n->Xn` forms a Hamilton cycle.
- `minof(Goal,Obj,Report)`
  Call `Report` each time a solution is found.
- Reification constraints

```
B #<=> (NX#=Y #¥/ NY#=X),
```

Planning

- Blocks world problem



Planning (Cont.)

- States and variables (m blocks and n states)

  $S_1\ S_2\ \ldots\ S_n$

  $S_i=(B_{i1},B_{i2},\ldots,B_{im})$

  B_{ij} = k (block j is on top of block k, block 0 means the table)

- Constraints
  - Every transition S_i -> S_{i+1} must be valid.

Channel Routing



N1={t(1),b(3)}
N2={b(1),t(2)}

Channel Routing (Cont.)

- Variables
  - For each net, use two variables L and T to represent the layer and track respectively
- Constraints
  - No two line segments can overlap
- Objective functions
  - Minimize the length (or areas) of wires

Protein Structure Predication

GPSQPTYPG
DDAPVEDLI
RFYDNLQQY
LNVVTRHRY

⇒



Protein Structure Predication (Cont.)

- Variables
  - Let $R=r_1,\ldots,r_n$ be a sequence of residues. A structure of R is represented by a sequence of points in a three-dimensional space $p_1,\ldots,p_n$ where $p_i=<x_i,y_i,z_i>$.
- Constraints
  - A structure forms a self-avoiding walk in the space
- The objective function
  - The energy is minimized

Constraint Solving Algorithms

- Generate and test
  - For each permutation of values for the variables, if the permutation satisfies the constraints then return it as a solution
- Backtracking
  - Begin with an empty partial solution. Keep extending the current partial solution by selecting an uninstantiated variable and assigning a value to it. If a partial solution violates one of the constraints, then backtrack.
- Propagation
  - Preprocess the constraints to make them consistent. Keep extending the current partial solution by selecting an uninstantiated variable and assigning a value to it. For each assignment, propagate the assignment to make the constraints consistent.

Constraint Propagation

X = Y+1 (X,Y in 1..5)

Algorithm	Change	Propagation
Preprocessing	generated	X in 2..5, Y in 1..4
Forward checking	X=3	X = 3, Y = 2
Interval consistency	5 notin X	X in 2..4, Y in 1..3
Arc consistency	4 notin X	X in [2,3,5], Y in [1,2,4]

Constraint Systems

- CLP systems
  - B-Prolog
  - BNR-Prolog
  - CHIP
  - CLP(R)
  - Eclipse
  - G-Prolog
  - IF/Prolog
  - Prolog-IV
  - SICStus
- Other systems
  - 2LP
  - ILOG solver
  - OPL
  - Oz
  - Gcode
  - Choco
- More information
  - Languages & compilers
  - Logic programming
  - Constraint programming